```
/* aml to summarise water quality data for a series of stations
/* based on the non-graphical part of BARCODE.
/* Michael Silberbauer IWQS DWAF June 1997
/* - minor changes                    November 2000
/* - fixed date problems with Arc 8.0 March 2001 (added option of YYYY-MM-DD too)
/* - added percentiles from Barcode   May 2001

/* Input specifications on command line:

&args CatPri CatSec CatTer Station VarType flow StnType Year1 Year2 Pcntl Debug
&sv Version = %AML$FULLFILE% v2.1
&type ************************************************************
&type Running %Version%
&if [null %CatPri%] or [quote %CatPri%] = #    and ~
     ( [quote %CatSec%] <> # or [quote %CatTer%]  <> # ) and ~
      [quote %Station%] <> #  &then
&do
 &type Usage: meaner CatPri CatSec CatTer Station\~
        VarType flow StnType Year1 Year2 Debug~
 \where:~
 \ CatPri limits the search to a particular primary catchment,~
 \            (e.g. C, Y for all or # to skip this selection).~
 \ CatSec  narrows the search to a secondary catchment,~
 \            if CatPri has been set.
 &type ~
 \ CatTer  further narrows the search if CatPri and CatSec are set~
 \            no data available yet so this option is ignored.~
 \ Station   selects only one station, e.g. C2R009Q01 -~
 \            use # # # for the three catch variables in this case.
 &type ~
 \ VarType   is general, salts, macro or trace,~
 \ Flow      is flow to plot flow (where available), or #,~
 \ StnType   is R or H for reservoir or hydro stn., or # for both,~
 \ Year1     is the starting year e.g. 1980~
 \ Year2     is the final year e.g 1996~
 \ Pcntl     percentile (= 90 percent by default)~
 \ Debug     is used by programmers to set debugging mode~
 \ ------------------------------------------------------------------~
 \ e.g. meaner # # # c9r002q01 macro flow R 1980 1996 ~
```

```
 \ e.g. meaner # # A23 # macro flow R 1999-10-01 2000-09-30 ~
 \ e.g. meaner C 2 # # macro flow H 1993 1996 95 ~
 \ DO NOT RUN MULTIPLE SESSIONS IN THE SAME DIRECTORY!
  &return

&end

&if [null %CatSec%] &then &sv CatSec = #
&if [null %CatTer%] &then &sv CatTer = #
&if [null %Station% ] &then &sv Station  = #
&sv CatPri  = [translate %CatPri%]
&sv All = .FALSE.
&if %CatPri% = Y &then
  &sv All = .TRUE.
&sv CatSec  = [translate %CatSec% ]
&sv CatTer  = [translate %CatTer% ]
&sv Station = [translate %Station%  ]
  &sv Single = .FALSE.
&if %Station% <> # and [length [quote %Station%]] > 1 &then
  &sv Single = .TRUE.
&if %Station% = # or [length [quote %Station%]] = 1 &then
  &sv CatQat = %Station%
&if [null %VarType% ] or [quote %VarType%] = # &then &sv VarType = Macro
&sv VarType = [substr [locase %VarType%] 1 1]
&sv Flow    = [substr [locase %Flow%] 1 1]
&if [null %StnType%] or [quote %StnType%] = '#' &then &sv StnType = B
&sv StnType = [translate %StnType%]
&if [null %Year1%] &then &sv Year1 = 1980
&if [null %Year2%] &then &sv Year2 = 1996

&if [null %Pcntl%] &then &sv Percentile = 90
  &else &sv Percentile = %Pcntl%
&if [substr [locase %Debug%] 1 1] = 'd' &then &sv Debug = .true.
&else &sv Debug = .false.

&type using: meaner %CatPri% %CatSec% %CatTer% %Station% %VarType% %Flow% %StnType% %Year1% %Year2% %Percentile%
Debug=%Debug%
&type ************************************************************
```

```
/* set defaults
&if [locase %VarType%] = p &then
&do
  &call SetAbbrev
  &sv Var1 = TP
  &sv nVariables = 1
  &sv ChemFile    = $WMSYS/wmdata/wq/inorganic.dat
&end

&if [locase %VarType%] = f &then
&do
  &call SetAbbrev
  &sv Var3 = Fluoride
  &sv Var2 = pH
  &sv Var1 = TDS
  &sv nVariables = 3
  &sv ChemFile    = $WMSYS/wmdata/wq/inorganic.dat
&end

&if [locase %VarType%] = g &then
&do
  &call SetAbbrev
  &sv Var3 = TAlkalinity
  &sv Var2 = pH
  &sv Var1 = TDS
  &sv nVariables = 3
  &sv ChemFile    = $WMSYS/wmdata/wq/inorganic.dat
&end

&if [locase %VarType%] = s &then
&do
  &call SetAbbrev
  &sv Var9 = Chloride
  &sv Var8 = Sulphate
  &sv Var7 = Potassium
  &sv Var6 = Magnesium
  &sv Var5 = Calcium
  &sv Var4 = Sodium
```

```
  &sv Var3 = TAlkalinity
  &sv Var2 = pH
  &sv Var1 = TDS
  &sv nVariables = 9
  &sv ChemFile    = $WMSYS/wmdata/wq/inorganic.dat
&end

&if [locase %VarType%] = m &then
&do
  &call SetAbbrev
  &sv Var20 = SAR
  &sv Var19 = Boron
  &sv Var18 = Conductivity
  &sv Var17 = pH
  &sv Var16 = TDS
  &sv Var15 = Calcium
  &sv Var14 = Magnesium
  &sv Var13 = Potassium
  &sv Var12 = Sodium
  &sv Var11 = TAlkalinity
  &sv Var10 = Chloride
  &sv Var9  = Fluoride
  &sv Var8  = Silica
  &sv Var7  = Sulphate
  &sv Var6  = NH4(N)
  &sv Var5  = NO3(N)
  &sv Var4  = KN
  &sv Var3  = PO4(P)
  &sv Var2  = TP
  &sv Var1  = Date
  &sv nVariables = 20
  /*&sv ChemFile    = /hri/db/raw/inorganic.tmp
  &sv ChemFile    = $WMSYS/wmdata/wq/inorganic.dat
/*&sv ChemFile    = /hri/db/raw/inorgc.dat
&end

&if [locase %VarType%] = t &then
&do
  &sv Var21 = Titanium
```

```
  &sv Var20 = Antimony
  &sv Var19 = Mercury
  &sv Var18 = Boron
  &sv Var17 = Beryllium
  &sv Var16 = Strontium
  &sv Var15 = Copper
  &sv Var14 = Vanadium
  &sv Var13 = Aluminium
  &sv Var12 = Molybdenum
  &sv Var11 = Chromium
  &sv Var10 = Iron
  &sv Var9  = Manganese
  &sv Var8  = Nickel
  &sv Var7  = Cadmium
  &sv Var6  = Lead
  &sv Var5  = Zinc
  &sv Var4  = Zirconium
  &sv Var3  = Barium
  &sv Var2  = Cobalt
  &sv Var1  = Arsenic
  &sv nVariables = 21
  &sv ChemFile    = /iwqs/prj/aqces/wq_data/trace.dat
&end

/* set file constants:

&sv StnPnt = /hri/db/cover/s-africa/nms_his
&sv Province    = /hri/db/cover/s-africa/spr_500
&sv Catch_Pri   = /hri/db/cover/s-africa/hca_1
&sv Catch_Sec   = /hri/db/cover/s-africa/hca_2
&sv Catch_Ter   = /hri/db/cover/s-africa/hca_3
&sv Catch_Qat   = /hri/db/cover/s-africa/hca_4
&sv Rivers      = /hri/db/cover/s-africa/wri_500
&sv RiversOut   = /hri/db/cover/s-africa/wri_not_sa
&sv Lakes       = /hri/db/cover/s-africa/wla_500
&sv Towns       = /hri/db/cover/s-africa/smu_500
&sv FlowDir     = /iwqs/db/flow/
&sv Portrait    = /prjws8/users/michael/aml/portrait.hp4
&sv Landscape   = /prjws8/users/michael/aml/landscape.hp4
```

```
&sv FormFeed    = /prjws8/users/michael/aml/formfeed.hp4
&sv User        = [translate [substr [username] 1 1]][substr [username] 2 [calc [length [username]] - 1]]
&sv PlotName    = mg[substr [username] 1 5]
&sv MnrTmpLst   = meanerlist.tmp
&sv MnrTmpSrt   = meaner.tmp
&sv MnrTmpXY    = meanerxy.tmp
&sv MnrPrjXY    = [before %MnrTmpXY% .].prj
&sv MnrTmpGeo   = meanergeo.tmp
&sv MnrPrjGeo   = [before %MnrTmpGeo% .].prj
&sv MnrScratch  = meanerscr.tmp
&sv MnrPrjScr   = [before %MnrScratch% .].prj

&call ISOdate
&sv RepFileID   = [subst [quote %ISOdate%] ' ' '']
&sv RepFileID   = [subst [quote %RepFileID%] '-' '']
&sv RepFileID   = [unquote [subst [quote %RepFileID%] ':' '']]
&sv Report      = /hri/db/barcode/mn[username]%RepFileID%.txt

&if [exists %Report% -file]    &then &sv delf [delete %Report% -file]
&if [exists %Report%r -file]   &then &sv delf [delete %Report%r -file]
&if [exists %MnrTmpXY% -file]  &then &sv delf [delete %MnrTmpXY% -file]
&if [exists %MnrTmpGeo% -file] &then &sv delf [delete %MnrTmpGeo% -file]
&if [exists %MnrPrjXY% -file]  &then &sv delf [delete %MnrPrjXY% -file]
&if [exists %MnrPrjGeo% -file] &then &sv delf [delete %MnrPrjGeo% -file]
&if [exists %MnrScratch% -file] &then &sv delf [delete %MnrScratch% -file]
&if [exists %MnrPrjScr% -file]  &then &sv delf [delete %MnrPrjScr% -file]
&sv closeall [close -all]

&sv ReportUnit  = [open %Report% openstatus -write]
&call list_variables

/* set date limits:
&if [length [quote [unquote %Year1%]]] = 10 &then &sv Date1 = %Year1%
&if [length [quote [unquote %Year1%]]] =  4 &then &sv Date1 = %Year1%-01-01
&if [length [quote [unquote %Year1%]]] =  2 &then &stop Y2K problem with date %Year1%
&if [length [quote [unquote %Year2%]]] = 10 &then &sv Date2 = %Year2%
&if [length [quote [unquote %Year2%]]] =  4 &then &sv Date2 = %Year2%-12-31
&if [length [quote [unquote %Year2%]]] =  2 &then &stop Y2K problem with date %Year2%
&sv DateRange = Date >= %Date1% & date <= %Date2%
```

```
/*date2day %Date1% .startday
/*date2day %Date2% .endday

&sv counter    = 0
&sv ReadStatus = 0
&sv Page       = 1

ARCPLOT
disp 1040
 temp

&type Searching for valid stations...
clearselect %StnPnt% point
&if %Single% &then reselect %StnPnt% point station = [quote %Station%]
&if %All% &then reselect %StnPnt% point station nc 'Z'
&if ^ %Single% and ^ %All% &then
&do
  reselect %StnPnt% point primary = [quote %CatPri%]
  &if [quote %CatSec%] <> # &then
    reselect %StnPnt% point secondary = [quote %CatPri%%CatSec%]
  &if [quote %CatTer%] <> # &then
    reselect %StnPnt% point tertiary  = [quote %CatPri%%CatSec%%CatTer%]
  &if [quote %CatQat%] <> # &then
    reselect %StnPnt% point quaternary = [quote %CatPri%%CatSec%%CatTer%%CatQat%]
  &if  [translate [quote %StnType%]] = 'R' or ~
       [translate [quote %StnType%]] = 'H' &then
  &do
    reselect %StnPnt% point stntype = [quote [translate [quote %StnType%]]]
    /*reselect %StnPnt% point SUBMONPT = '01'
  &end
&end
&if ^ %Single% and %All% &then
&do
  &if  [translate [quote %StnType%]] = 'R' or ~
       [translate [quote %StnType%]] = 'H' &then
  &do
    reselect %StnPnt% point stntype = [quote [translate [quote %StnType%]]]
    /*reselect %StnPnt% point SUBMONPT = '01'
  &end
```

```
&end

&sv nStations = [before [show select %StnPnt% point] ,]
infofile %StnPnt% point %MnrTmpLst% primary secondary tertiary quaternary station init
&data ARC INFO
  ARC
  SELECT [translate [entryname %MnrTmpLst%]]
  SORT ON QUATERNARY, STATION
  Q STOP
&end

list %MnrTmpLst% info 1 10
&type Number of stations selected = %nStations%
clearselect %MnrTmpLst% info
reselect    %MnrTmpLst% info station <> ''

&if %nStations% = 0 &then
&do
  QUIT
  &stop No stations found...
&end

&do n = 1 &to %nStations%
   &if [exists %MnrTmpXY%   -file] &then &ty [delete %MnrTmpXY%   -file]
   &if [exists %MnrTmpGeo%  -file] &then &ty [delete %MnrTmpGeo%  -file]
   &if [exists %MnrPrjXY%   -file] &then &ty [delete %MnrPrjXY%   -file]
   &if [exists %MnrPrjGeo%  -file] &then &ty [delete %MnrPrjGeo%  -file]
   &if [exists %MnrScratch% -file] &then &ty [delete %MnrScratch% -file]
   &if [exists %MnrPrjScr% -file]  &then &ty [delete %MnrPrjScr% -file]
   &sv PrjUnit     = [open %MnrTmpXY% openstatus -write]

   &if [show program] <> ARCPLOT &then ARCPLOT

   &sv Station     = [show select %MnrTmpLst% info %n% item station]
   clearselect %StnPnt% point
   reselect    %StnPnt% point station = [quote %Station%]
   &sv Description = [show select %StnPnt% point 1 item description]
   &sv mx          = [show select %StnPnt% point 1 item x-coord]
   &sv my          = [show select %StnPnt% point 1 item y-coord]
```

```
   &sv fLatitude   = [show select %StnPnt% point 1 item latitude]
   &sv fLongitude  = [show select %StnPnt% point 1 item longitude]
   &sv Primary     = [show select %StnPnt% point 1 item primary]
   &sv Secondary   = [show select %StnPnt% point 1 item secondary]
   &sv Tertiary    = [show select %StnPnt% point 1 item tertiary]
   &sv Quaternary  = [show select %StnPnt% point 1 item quaternary]
   &sv wp [write %PrjUnit% [quote %mx% %my%]]
   &sv cl [close %PrjUnit%]
   &call AlbGeo

   /* remove funny characters from description:
   &sv description [quote [subst [quote %description%] '''' ` ]]
   &sv description [subst %description% '[' '-']
   &sv description [subst %description% ']' '-']
   &sv description [subst %description% ',' '-']
   &sv description [subst %description% '``S' '`s']
   /* remove derogatory place names:
   &sv description [subst %description% 'Kaffer' 'XX']
   &sv description [subst %description% 'Kaffir' 'XX']
   /* remove quotes and back-quotes:
   &sv description [unquote %description%]
   /*&sv description [substr %description% 2 [calc [length %description%] - 2] ]
   &sv description [unquote %description%]

   &type (%n% of %nStations%) %Station% %Longitude% %Latitude% - %Description%
   &sv PlotName    = %Station%

   &sv nVars  = %nVariables%
   &if %Flow% = f &then &type don't call FlowCalc

   &call CalcLoop

   &if [show program] = ARCPLOT &then QUIT /* ARCPLOT (so as to close .gra file)

&end

/*&if %Debug% &then &message &on
&message &on
```

```
&sv wr =~
   [write %ReportUnit% [quote %Date1% to %Date2% from IWQS GIS file %ChemFile% to %Report%]]
&call datestamp
&sv CloseReport = [close -all]

/*&if [exists [locase %Report%]r -file] &then \rm [locase %Report%]r
/*cat %Landscape% [locase %Report%] %Portrait% > [locase %Report%]r
/*unix2dos %Report%r %Report%r

/*&if %Display% = p &then lp -dhp4si %Report%r
/*&if %Display% = p &then &sys pr -f -w120 -l66 [locase %Report%r] | lp -dhp4si
&type Results are summarised in file %Report%
&type Combine files with a command of the form:
&type grep -v -h file *txt | grep -v Data | grep -v ":" | grep -v "usage meaner" > all.txt

&type End of station list...

&if [show program] = ARCPLOT &then QUIT /* arcplot session

&if [exists %MnrTmpXY% -file]  &then &sv delf [delete %MnrTmpXY% -file]
&if [exists %MnrTmpGeo% -file] &then &sv delf [delete %MnrTmpGeo% -file]
&if [exists %MnrPrjXY% -file]  &then &sv delf [delete %MnrPrjXY% -file]
&if [exists %MnrPrjGeo% -file] &then &sv delf [delete %MnrPrjGeo% -file]
&if [exists %MnrScratch% -file] &then &sv delf [delete %MnrScratch% -file]
&if [exists %MnrPrjScr% -file]  &then &sv delf [delete %MnrPrjScr% -file]

&return

/* - - - - - - - - - - - - - - - - - -

&routine CalcLoop
/* routine to calculate statistics:

&sv counter = %counter% + 1

/*&if %counter% > 2 &then &stop Finished - test mode.

&sv closeall [close -all]
&sv ReportUnit  = [open %Report% openstatus -append]
```

```
&if %Debug% &then &echo &on

&sv NoData_AtAll = .TRUE.
&type Calculating summary statistics...

&sv ValList = [unquote '']

&do n = 1 &to %nVars%
  &call StatChem
&end

&if ^ %NoData_AtAll% &then &sv wr =~
    [write %ReportUnit% [quote %Station%,%Longitude%,%Latitude%%ValList%,%Description%]]

&return

/* - - - - - - - - - - - - - - - - - -

&routine datestamp

&call ISOdate

&sv wr =~
   [write %ReportUnit% [quote usage meaner %CatPri% %CatSec% %CatTer% %Station% %VarType% %Flow% %StnType% %Year1%
%Year2% %Percentile% Debug=%Debug%]]
&sv Address = Institute for Water Quality Studies\Department of Water Affairs & Forestry\Private Bag X313
PRETORIA\tel (012) 808 0375\fax (012) 808 2702\E-mail: eck@dwaf-hri.pwv.gov.za
&sv wr =~
   [write %ReportUnit% [quote %Version% data summary by user [username] on %ISOdate%]]
&sv wr =~
   [write %ReportUnit% [quote %Address%]]

&return

/* - - - - - - - - - - - - - - - - - -

&routine list_variables
```

```
&sv VarList = Station,Longitude,Latitude,
&do n = 1 &to %nVariables%
   /*&sv VarList = %VarList%n[value Var%n%],
   /*&sv VarList = %VarList%min[value Var%n%],
   /*&sv VarList = %VarList%med[value Var%n%],
   /*&sv VarList = %VarList%p%Percentile%[value Var%n%],
   /*&sv VarList = %VarList%max[value Var%n%],
   &sv VarList = %VarList%n[value VarAbbr%n%],
   &sv VarList = %VarList%min[value VarAbbr%n%],
   &sv VarList = %VarList%med[value VarAbbr%n%],
   &sv VarList = %VarList%p%Percentile%[value VarAbbr%n%],
   &sv VarList = %VarList%max[value VarAbbr%n%],
&end
&sv VarList = %VarList%Description
&sv wr =~
   [write %ReportUnit% [quote %VarList%]]

&return

/* - - - - - - - - - - - - - - - - - -
/* this routine is unused at the moment...
&routine FlowCalc

&sv nVars = %nVars% + 1
&sv Var%nVars% = Flow
&sv FlowStn = [substr %Station% 1 6]A01
&if [exists FLOWMNR.TMP -info] &then &sv deli [delete FLOWMNR.TMP -info]
&if [exists %FlowDir%%FlowStn% -file] &then
&do

  &DATA arc info
     ARC
     DFMT DMY-/
     DEFINE FLOWMNR.TMP
     DATE   ,  8,11,D
     FLOW   ,  4, 9,F,3
     CODE   ,  1, 1, C
     NDATE  ,  4,10,F,0
```

```
     ADD FROM %FlowDir%%FlowStn%
     CALC NDATE = DATE
  Q STOP
  &end

&end

&else
  &type Sorry - no flow data for station %FlowStn%
&return

/* - - - - - - - - - - - - - - - - - -

&routine StatChem

&sv yMin = -9
&sv yMdn = -9
&sv yPct = -9
&sv yMax = -9
&if [locase [value Var%n%]] <> flow &then
&do
  &if ^ %Debug% &then &message &off
  /*&type [value Var%n%]
  /*&echo &on
  CLEARSELECT %ChemFile% INFO
  RESELECT    %ChemFile% INFO STATION = [quote %Station%]
  &if [locase [value Var%n%]] <> date &then
    RESELECT    %ChemFile% INFO [value Var%n%] > 0
  RESELECT    %ChemFile% INFO %DateRange%
  &sv ndata = [before [show select %ChemFile% info] ,]
&type %nData%
  &if %ndata% > 0 &then
  &do
    infofile %ChemFile% INFO %MnrTmpSrt% [translate [value Var%n%]] init
    &call Median
    &call Percentile
    &sv NoData_AtAll = .FALSE.

    &sv yMin = %Minimum%
```

```
    &sv yMdn = %Median%
    &sv yPct = %PcntVar%
    &sv yMax = %Maximum%
    &if [locase [value Var%n%]] <> date &then
    &do
      &sv yMin = [calc [round [calc 1000 * %yMin%]] / 1000]
      &sv yMdn = [calc [round [calc 1000 * %yMdn%]] / 1000]
      &sv yPct = [calc [round [calc 1000 * %yPct%]] / 1000]
      &sv yMax = [calc [round [calc 1000 * %yMax%]] / 1000]
    &end
    &if %Debug% &then &message &on
  &end
  &else &if %ndata% = 0 &then &type No data.

  &type %Station% [value Var%n%]:  n = %ndata%, Minimum = %yMin%, median = %yMdn%, percentile%Percentile% = %yPct%
and maximum = %yMax%
  &sv ValList = %ValList%,%ndata%,%yMin%,%yMdn%,%yPct%,%yMax%
  &sv dsf = [delete %MnrTmpSrt% -info]
&end
&else
   &type Program error - should not try to plot flow here...

&return

/* - - - - - - - - - - - - - - - - - -

&routine StatDate

&sv yMin = -9
&sv yMdn = -9
&sv yMax = -9
&if [locase [value Var%n%]] = date &then
&do
  &if ^ %Debug% &then &message &off
  CLEARSELECT %ChemFile% INFO
  RESELECT    %ChemFile% INFO STATION = [quote %Station%]
  RESELECT    %ChemFile% INFO [value Var%n%] > 0
  RESELECT    %ChemFile% INFO %DateRange%
  &sv ndata = [before [show select %ChemFile% info] ,]
```

```
  &if %ndata% > 0 &then
  &do
    infofile %ChemFile% INFO %MnrTmpSrt% [value Var%n%]
    &sv Percentile = 50
    &sv PcntRec    = [round [max 1 [calc %Percentile% / 100 * %ndata%]]]
    &if %PcntRec% > %ndata% &then &sv PcntRec = %ndata%
    &sv NoData_AtAll = .FALSE.

    &data ARC INFO
      ARC
      SELECT [translate %MnrTmpSrt%]
      SORT ON [translate [value Var%n%]]
      Q STOP
    quit
    &end

    &sv yMin = [show select %MnrTmpSrt% info 1         item [quote [value Var%n%]]]
    &sv yMdn = [show select %MnrTmpSrt% info %PcntRec% item [quote [value Var%n%]]]
    &sv yMax = [show select %MnrTmpSrt% info %ndata%   item [quote [value Var%n%]]]
    &if %Debug% &then &message &on
  &end
  &else &if %ndata% = 0 &then &type No data.

  &type %Station% [value Var%n%]:  n = %ndata%, Minimum = %yMin%, median = %yMdn% and maximum
= %yMax%
  &sv ValList = %ValList%, %ndata%,%yMin%,%yMdn%,%yMax%
  &sv dsf = [delete %MnrTmpSrt% -info]
&end
&else
   &type Program error - wrong variable in StatDate

&return


/* - - - - - - - - - - - - - - - - - -

&routine StatFlow /* not used, and needs updating to new mean and median...

&if [exists FLOWMNR.TMP -info] &then
```

```
&do
   CLEARSELECT FLOWMNR.TMP INFO
   RESELECT    FLOWMNR.TMP INFO FLOW >= 0
   RESELECT    FLOWMNR.TMP INFO %DateRange%
   &sv ndata = [before [show select FLOWMNR.TMP info] ,]
   &if %ndata% > 0 &then
/*** Must put in median calculation as above ***
   &do
     &sv Percentile = 50
     &sv PcntRec    = [round [max 1 [calc %Percentile% / 100 * %ndata%]]]
     &if %PcntRec% > %ndata% &then &sv %PcntRec% = %ndata%
     &sv NoData_AtAll = .FALSE.

     &data ARC INFO
       ARC
       SELECT FLOWMNR.TMP
       SORT ON FLOW
       Q STOP
     quit
     &end

     &sv yMin = [show select FLOWMNR.TMP info 1         item FLOW]
     &sv yMdn = [show select FLOWMNR.TMP info %PcntRec% item FLOW]
     &sv yMax = [show select FLOWMNR.TMP info %ndata%   item FLOW]
     &sv yMin = [calc [round [calc 10000 * %yMin%]] / 10000]
     &sv yMdn = [calc [round [calc 10000 * %yMdn%]] / 10000]
     &sv yMax = [calc [round [calc 10000 * %yMax%]] / 10000]
     &type %Station% [value Var%n%]:  n = %ndata%, Minimum = %yMin%, median = %yMdn% and maximum = %yMax%
     &sv ValList = %ValList%, %ndata%,%yMin%,%yMdn%,%yMax%
     &sv dsf = [delete %MnrTmpSrt% -info]
     &if %Debug% &then &message &on
   &end
   &else &type No flow data available in %FlowStn% for %DateRange%
&end
&else &type Sorry - no flow data file %FlowStn% in %FlowDir%

&return

/* - - - - - - - - - - - - - - - - -
```

```
&routine SetAbbrev

&sv VarAbbr22 = xx
&sv VarAbbr21 = x
&sv VarAbbr20 = SAR
&sv VarAbbr19 = B
&sv VarAbbr18 = EC
&sv VarAbbr17 = pH
&sv VarAbbr16 = TDS
&sv VarAbbr15 = Ca
&sv VarAbbr14 = Mg
&sv VarAbbr13 = K
&sv VarAbbr12 = Na
&sv VarAbbr11 = TAL
&sv VarAbbr10 = Cl
&sv VarAbbr9  = F
&sv VarAbbr8  = Si
&sv VarAbbr7  = SO4
&sv VarAbbr6  = NH4(N)
&sv VarAbbr5  = NO3(N)
&sv VarAbbr4  = KN
&sv VarAbbr3  = PO4(P)
&sv VarAbbr2  = TP
&sv VarAbbr1  = Date

&return

/* - - - - - - - - - - - - - - - - - -

&routine AlbGeo

&if [exists %MnrScratch% -file] &then &ty [delete %MnrScratch% -file]
&if [exists %MnrPrjScr% -file]  &then &ty [delete %MnrPrjScr% -file]
&if [exists %MnrPrjXY%   -file] &then &ty [delete %MnrPrjXY%   -file]
&if [exists %MnrPrjGeo%  -file] &then &ty [delete %MnrPrjGeo%  -file]
&data arc
  project file %MnrTmpXY% %MnrScratch%
  input
```

```
  projection albers
  units meters
  spheroid clarke1880
  parameters
  -18 0 0
  -32 0 0
  24 0 0
  00 0 0
  0.0
  0.0
  output
  projection geographic
  units dd
  spheroid clarke1880
  parameters
  end
quit
&end

&sv GeoUnit [open %MnrScratch% openstatus -read]
&sv GeoCoord [read %GeoUnit% readstatus]
&sv Longitude = [extract 1 [unquote %GeoCoord%]]
&sv Latitude  = [extract 2 [unquote %GeoCoord%]]

&return

/*-------------------------------------------------------------------------
&routine Percentile

/* Calculate nth percentile.  Algorithm supplied by John Carter.
/* Calculate the record, then sort the data for each variable and select the nth percentile.
/* First calculate the (possibly) theoretical ExactRecord for the percentile,
/* then find the two integers surrounding this value:

&sv PcntRec = [round [max 1 [calc %Percentile% / 100 * %ndata%]]]
&if %PcntRec% > %ndata% &then &sv PcntRec = %ndata%
&sv ExactRec = [calc ( %Percentile% * %ndata% ) / 100 ]
&sv Recd1      [truncate %ExactRec%]
&if %Recd1% < 1 &then &sv Recd1 = 1
```

```
&sv Recd2    = [calc %Recd1% + 1]
&if %Recd2% > %ndata% &then &sv Recd2 = %ndata%

&data ARC INFO
  ARC
  SELECT [translate [entryname %MnrTmpSrt%]]
  SORT ON [translate [value Var%n%]]
  Q STOP
 q
&end

&sv FirstValue  = [show select %MnrTmpSrt% INFO %Recd1% item [value Var%n%]]
&sv SecondValue = [show select %MnrTmpSrt% INFO %Recd2% item [value Var%n%]]
&if [locase [value Var%n%]] = date &then
&do
  &sv PcntVar  = [show select %MnrTmpSrt% info %Recd1% item [value Var%n%]]
  &return
&end

/* Interpolate between these two records, by the cunning method of finding out
/* the exact percentile of the two records surrounding the percentile we want:

&sv Val2-1    = [calc %SecondValue% - %FirstValue%]
&sv Pcntile1  = [calc ( %Recd1% / %ndata% ) * 100]
&sv Pcntile2  = [calc ( %Recd2% / %ndata% ) * 100]
&if %Pcntile1% <> %Pcntile2% &then
&do
  &sv PcntRatio = [calc ( %Percentile% - %Pcntile1% ) / ( %Pcntile2% - %Pcntile1% ) ]
  &sv PcntVar   = [calc ( %PcntRatio% * %Val2-1% ) + %FirstValue%]
&end
  &else &sv PcntVar = %FirstValue%

&return

/*-----------------------------------------------------------------------
&routine Median

/* calculate median:
```

```
&if %ndata% = 1 &then
&do
  &sv Median = [show select %ChemFile% info 1 item [value Var%n%]]
  &sv Minimum = [show select %MnrTmpSrt% info 1         item [value Var%n%]]
  &sv Maximum = [show select %MnrTmpSrt% info %ndata%   item [value Var%n%]]
  &return
&end

&data ARC INFO
  ARC
  SELECT [translate [entryname %MnrTmpSrt%]]
  SORT ON [translate [value Var%n%]]
  Q STOP
 q
&end
&sv Minimum = [show select %MnrTmpSrt% info 1         item [value Var%n%]]
&sv Maximum = [show select %MnrTmpSrt% info %ndata%   item [value Var%n%]]

&sv MedRec = [round [max 1 [calc 0.5 * %ndata%] ] ]
&if %MedRec% > %ndata% &then &sv MedRec = %ndata%
&if [locase [value Var%n%]] = date &then
&do
  &sv Median  = [show select %MnrTmpSrt% info %MedRec%  item [value Var%n%]]
  &return
&end
&if [mod %ndata% 2] = 0 &then    /* even
&do
  &sv MedRec2 = %MedRec% + 1
  &sv Median1 = [show select %MnrTmpSrt% info %MedRec%  item [value Var%n%]]
  &sv Median2 = [show select %MnrTmpSrt% info %MedRec2% item [value Var%n%]]
  &sv Median  = [calc ( %Median1% + %Median2% ) / 2 ]
&end
&else                             /* odd
&do
  &sv Median  = [show select %MnrTmpSrt% info %MedRec%  item [value Var%n%]]
&end

&return
```

```
/*-------------------------------------------------------------------------
&routine ISOdate

/* date in ISO format
&sv ISOdate [date -year]-[substr[date -usa] 1 2]-[substr[date -vmsdate] 1 2] - [before[date -vmstime] .]

&return


/*-------------------------------------------------------------------------
```